

ATARI®



**INITIATION A LA
PROGRAMMATION
EN BASIC (cours n° 3)**

© 1981 ATARI, Inc.

ATARIINSIDE



INITIATION A LA PROGRAMMATION EN BASIC (cours n° 3)

© 1981 ATARI, Inc.

Copyright et copie :

A l'achat de ce programme pour ordinateur et de sa documentation associée (le logiciel), vous obtenez le droit d'utiliser ce logiciel pour votre usage personnel seulement sans en effectuer de copies. Ce logiciel est déposé. Il vous est interdit de le reproduire, le traduire ou le dupliquer sans l'autorisation écrite d'Atari Inc.

GARANTIE

Conformément à la loi, la présente vente est soumise à la garantie légale des défauts et vices cachés.

Nous garantissons, pendant 30 jours après la date d'achat, que le support sur lequel ce programme ATARI® est enregistré, ne comporte aucun défaut.

Si toutefois, il se trouvait que ce programme ne puisse se charger normalement, et que la disquette, la cassette ou la cartouche ci-incluses en soit la cause, veuillez le rapporter à votre revendeur avec une preuve d'achat datée, afin qu'il puisse appliquer la garantie qui se borne strictement à l'échange de ce programme par un autre identique, dans les meilleurs délais.

Cette garantie ne s'applique plus dès lors que le support montre des signes évidents et anormaux d'usure, de contraintes mécaniques (pliures, déformations dues à la chaleur, froissage de la bande, tentative de démontage ou d'ouverture du support, etc.) ou de mauvaise utilisation ou détérioration (renversement d'un liquide sur le support, empreinte de doigts sur les parties magnétiques, altération électromagnétique, etc.). La garantie est aussi exclue si ce produit n'est pas d'origine ATARI ou s'il a été modifié par quiconque autre que par les techniciens ou ingénieurs d'ATARI.

L'acheteur est tenu, dès son acquisition, de mettre à l'épreuve le logiciel de ce programme ATARI, de vérifier la véracité de ses résultats et de signaler sur le champ toute anomalie éventuelle à son vendeur afin que celui-ci puisse en faire vérifier l'exactitude par ATARI en le retournant pour son remplacement, dans les meilleurs délais.

1 - INTRODUCTION	1
-------------------------	----------

2 - COMMENT UTILISER CE COURS	2
--------------------------------------	----------

Matériel nécessaire	3
Instructions générales pour le chargement d'une cassette	3
Préparation aux exercices pratiques	4

3 - SOMMAIRE DES COURS DE LA CASSETTE A INTRODUCTIONS AUX EFFETS SONORES	7
---	----------

4 - DESCRIPTION DES DIFFERENTES PARTIES DE LA CASSETTE A	
---	--

Leçon 1 : Registres sons	9
Leçon 2 : Effets sonores	9
Leçon 3 : Programmes d'effets sonores	10
Leçon 4 : Programmes plus efficaces	13
Leçon 5 : Gammes musicales	16
Leçon 6 : Identifions les notes	18
Leçon 7 : Effets sonores à plusieurs voix	19
Questionnaire final: Les sons	20

5 - SOMMAIRE DES COURS DE LA CASSETTE B INTRODUCTION AUX GRAPHIQUES	23
--	-----------

6 - DESCRIPTION DES DIFFERENTES PARTIES DE LA CASSETTE B	
---	--

Leçon 1 : Démonstrations graphiques	25
Leçon 2 : Changer de couleur	27
Leçon 3 : Modes graphiques 1 et 2	28
Leçon 4 : De vrais graphiques	30
Leçon 5 : Quelques astuces	33
Leçon 6 : Position et animation	35
Leçon 7 : Caractères graphiques	37
Questionnaire final: Les graphiques	38

INTRODUCTION

Le programme INITIATION A LA PROGRAMMATION EN BASIC III décrit les techniques de programmation en langage BASIC. Ce cours se divise en deux parties. Chaque leçon utilise une présentation interactive vous permettant d'expérimenter immédiatement les notions que vous venez d'apprendre. La première cassette de ce cours, la cassette A, vous donne les connaissances de base qui vous sont nécessaires pour créer des effets sonores. Vous pouvez ensuite passer à la cassette B qui vous permettra de réaliser des graphiques et des animations à l'écran.

Ce logiciel comprend le matériel suivant :

- La cassette A
- La cassette B
- Ce livre de travail servant aux deux cassettes.

COMMENT UTILISER CE COURS

MATERIEL NECESSAIRE

- Ordinateur ATARI 600XL ou 800XL
- ou
- Ordinateur ATARI 400 ou 800 équipé de la cartouche de langage ATARI BASIC.
 - Magnétocassette ATARI 1010 ou ATARI 410.
 - Cassettes A et B de ce cours.

INSTRUCTIONS GENERALES POUR LE CHARGEMENT DE CE LOGICIEL EN CASSETTE

1. Connectez l'ordinateur ATARI au récepteur de télévision et alimentez-le par l'intermédiaire de son transformateur. Suivez pour cela les indications données dans le manuel d'utilisation de l'ordinateur.
2. Raccordez le magnétocassette ATARI à l'ordinateur puis à une source d'alimentation.

NOTE : Si vous avez connecté en série d'autres périphériques ATARI à votre console d'ordinateur et ne voulez pas les déconnecter, reliez votre magnétocassette ATARI au connecteur entrée/sortie de la dernière unité de la chaîne.

Si une unité de disquette ATARI est connectée à l'ordinateur, le Système d'Exploitation de la Disquette utilise une partie non négligeable de la mémoire de l'ordinateur. En conséquence, il se pourrait que dans le cas de l'utilisation d'un ordinateur avec 16 ko de mémoire vive, il n'y ait plus assez de place en mémoire pour faire tourner certains programmes de ce cours. Dans ce cas, n'hésitez pas à déconnecter l'unité de disquette, éteignez votre ordinateur, rallumez-le et recommencez la procédure de chargement du programme.

3. Si vous possédez un ordinateur ATARI 400 ou ATARI 800, ouvrez la porte placée sur le dessus de l'ordinateur, et insérez la cartouche BASIC correctement. Fermez la porte à fond.
4. Mettez sous tension votre téléviseur. Le réglage du volume sonore ne doit pas être à zéro car la voix enregistrée sur ces cassettes est reproduite par le haut-parleur de votre téléviseur.
5. Mettez sous tension votre ordinateur.
6. Si tout est correctement branché et mis sous tension, vous devez voir apparaître le message **READY** sur l'écran de télévision.
7. En appuyant sur la touche **STOP/EJECT** de votre magnétocassette, vous ouvrez la porte de celui-ci.
8. Placez alors la cassette. Vous devez pouvoir lire l'étiquette de la face que vous voulez utiliser.

9. Fermez la porte et si nécessaire, rembobinez la bande à son début. Mettez alors le compteur à zéro.
10. Tapez **CLOAD** sur le clavier de l'ordinateur et appuyez sur **RETURN**. Vous entendrez alors un bip sonore vous rappelant qu'il faut appuyer sur la touche **PLAY** du magnétocassette.
11. Appuyez sur la touche **PLAY** du magnétocassette puis sur la touche **RETURN** du clavier de l'ordinateur. Le magnétocassette doit se mettre alors à fonctionner. Les bips sonores et sons divers que vous entendez dans le haut-parleur de votre téléviseur vous indiquent simplement que le programme de la leçon 1 est en train de se charger en mémoire.

NOTE : Si vous rencontrez des problèmes de chargement, c'est-à-dire si vous voyez apparaître des messages d'erreurs à l'écran, déconnectez tous les éléments périphériques et raccordez le magnétocassette directement à la fiche latérale de votre ordinateur. D'autre part, il est possible que la bande amorce de la cassette soit trop longue. Si une erreur se produit sur le premier programme de la face de la cassette, bobinez alors à la main jusqu'à faire apparaître le début de la bande dans les fenêtres de la cassette. En passant ainsi la bande amorce, votre programme devrait se charger normalement.

12. Lorsque vous voyez apparaître sur l'écran de votre ordinateur un nouveau message **READY**, le programme s'est correctement chargé dans la mémoire de l'ordinateur. Laissez bien la touche **PLAY** du magnétocassette enfoncée. Relevez alors le numéro du compteur et placez-le dans le tableau que nous avons prévu à cet effet. Cela vous permettra par la suite de revenir plus facilement au début de la voix ou au début du programme. Ce nombre, dans notre exemple, correspond à la colonne « Début de la voix ». Si vous désirez recommencer cette leçon, rembobinez simplement la bande jusqu'à ce numéro et reprenez la procédure à partir du point ci-dessous.
13. Tapez **RUN** et appuyez sur la touche **RETURN** pour démarrer la leçon 1. Dès à présent, suivez les instructions données dans le cours. Arrivé à la fin de la leçon 1, notez à nouveau le numéro du compteur du magnétocassette et placez-le dans le tableau, à la colonne « Fin de la voix ».
14. Ne rembobinez pas la cassette. Lorsque vous êtes prêt pour la leçon 2, suivez la même procédure que celle que nous venons de voir à partir du point 10.

Si vous désirez étudier une leçon particulière de ce cours, rembobinez complètement la bande, placez le compteur à 000 puis avancez la bande jusqu'à ce que le compteur arrive au numéro que vous aurez reporté dans la colonne « Début du programme ». Reprenez au point 10 ci-dessus.

Pour recommencer une leçon que vous venez juste de faire, rembobinez la bande jusqu'à ce que le compteur arrive au numéro que vous avez placé dans la colonne « Début de la voix ».

NOTE : La touche d'inversion vidéo (ou touche logo ATARI) dont parle le speaker est représentée sur les ordinateurs de la série XL par une touche comportant un rectangle blanc et noir. Sur les ordinateurs ATARI 400 ou 800, le dessin de cette touche représente le logo ATARI.

REMARQUES GENERALES

La mémoire de l'ordinateur peut être divisée en deux parties : mémoire vive et mémoire morte.

La mémoire morte contient des informations placées par le constructeur. Ces informations sont inaltérables et permettent à l'ordinateur de fonctionner.

La mémoire vive stocke temporairement des programmes et des données dans votre ordinateur. C'est dans cette partie de la mémoire que vous chargez votre programme, que s'enregistrent les caractères que vous tapez au clavier, que vous créez de nouveaux programmes etc.

Le fait d'éteindre l'ordinateur efface toutes les informations contenues dans la mémoire vive. Il existe une instruction qui permet d'effacer la mémoire vive : il s'agit de l'instruction **NEW**. Après avoir tapé cette instruction, il faut appuyer sur la touche **RETURN** pour que l'ordinateur la prenne en considération et l'exécute.

L'instruction **CLOAD** permet le transfert d'un programme contenu sur une cassette dans la mémoire de votre ordinateur. Avant de taper cette instruction, si vous avez déjà utilisé votre ordinateur depuis qu'il est mis sous tension, vous devez effacer sa mémoire en tapant l'instruction **NEW**.

Parvenu à la fin d'un cours, si vous désirez réaliser vous-même des exemples et des applications, vous devrez également taper **NEW** et appuyer sur la touche **RETURN** pour enlever de la mémoire le programme de cours lui-même.

SOMMAIRE DES COURS DE LA CASSETTE A: INTRODUCTION AUX EFFETS SONORES

- Leçon 1: **REGISTRES SONS.**
Décrit les diverses caractéristiques d'un son généré par ordinateur. Vous changez chaque caractéristique puis vous écoutez le résultat.
- Leçon 2: **EFFETS SONORES.**
Explique les principes de base des programmes générateurs de sons simples comme par exemple des explosions. Vous apprenez également à imiter des instruments de musique.
- Leçon 3: **PROGRAMMES D'EFFETS SONORES.**
Passe en revue les différentes étapes logiques servant dans l'élaboration d'un programme permettant de jouer de la musique.
- Leçon 4: **PROGRAMMES PLUS EFFICACES.**
Introduction des instructions **DATA** ainsi que d'autres techniques permettant d'améliorer les programmes développés dans le cours 3.
- Leçon 5: **GAMMES MUSICALES.**
Décrit les principes de base d'écriture de la musique.
- Leçon 6: **IDENTIFIONS LES NOTES.**
Divers sons musicaux sont joués en même temps qu'affichés sur l'écran. Vous identifiez les sons et leurs valeurs. A la fin du cours, vous traduisez une portée musicale en programme compréhensible par votre ordinateur.
- Leçon 7: **EFFETS SONORES A PLUSIEURS VOIX.**
Montre comment créer des programmes de musique et des effets sonores utilisant plusieurs voix.
- QUESTIONNAIRE FINAL: Ensemble de questions passant en revue les différentes parties de cette cassette.

Cassette A	Leçons	Valeurs au compteur du magnétocassette		
		Début du chargement	Début de la voix	Fin de la voix
Face 1	1	000	145	156
	2	156	187	289
	3	289	375	415
	4	415	439	
Face 2	5	000	39	159
	6	159	192	265
	7	265	292	371
	Test final	371		

Figure 2 - Valeurs du compteur du magnétocassette pour les programmes de la cassette A

DESCRIPTION DES DIFFERENTES PARTIES DE LA CASSETTE A

LEÇON 1 : REGISTRES SONS

§ 1 - Le registre son

Trois variables déterminent la qualité d'un son produit par un registre son: la hauteur du son (ou tonalité), le timbre et l'intensité (ou volume) sonore.

- La hauteur du son est représentée sur les ordinateurs ATARI par un nombre compris entre 1 et 255. Plus cette valeur numérique est élevée, plus le son est grave. Reportez-vous au paragraphe 19 pour obtenir l'équivalence entre les valeurs numériques et les hauteurs de notes normalisées.
- Le timbre caractérise la pureté du son. Elle est représentée par une valeur numérique variant entre 0 et 15. Une valeur de 10 correspond à un son pur tandis qu'une valeur 12 donnera une sorte de ronflement, plus particulièrement avec une hauteur de note de 200.
- Le dernier paramètre contrôle le volume sonore. La valeur numérique correspondante varie de 1 à 15, une valeur élevée correspondant à une intensité sonore forte. Une valeur de 8 correspond à la moyenne généralement utilisée.

§ 2 - Les quatre registres son

L'ordinateur ATARI comprend quatre registres son numérotés de 0 à 3.

§ 3 - L'instruction SOUND

Pour produire un son, vous devez utiliser l'instruction SOUND avec quatre paramètres: le numéro du registre utilisé, la hauteur de la note, le timbre et l'intensité sonore. Des virgules doivent séparer ces différents paramètres.

Exemple:

```
SOUND 0,100,10,12
```

- 0 correspond au premier registre son,
- 100 à la hauteur de la note,
- 10 au timbre,
- 12 à l'intensité sonore.

§ 4 - Comment arrêter un son

Bien sûr, vous devez avoir réglé votre téléviseur de manière à entendre le résultat de vos instructions. Après avoir exécuté une instruction SOUND, l'ordinateur continue de générer ce son jusqu'à ce qu'un des événements suivants se produise:

- Vous utilisez une nouvelle instruction SOUND avec la valeur d'intensité sonore égale à 0.
- L'ordinateur exécute une instruction END. Par exemple:

```
10 SOUND 1,50,10,8
20 FOR X=1 TO 1000:NEXT X
30 END
```
- Vous appuyez sur la touche **SYSTEM RESET**.
- Vous baissez l'intensité du son sur votre téléviseur.

§ 5 - Application pratique de la leçon 1

Note: Effacez la mémoire de l'ordinateur avant chaque exercice pratique en tapant **NEW** et en appuyant sur **RETURN**.

1. Tapez cette instruction SOUND puis appuyez sur **RETURN**.

```
SOUND 0,50,2,8
```

Pour arrêter ce son, tapez **END**.

2. Changez la hauteur du son dans l'instruction précédente. Essayez plusieurs fois jusqu'à obtenir une bonne idée de la gamme utilisable. Reportez-vous au paragraphe 22 pour obtenir les équivalences entre les valeurs numériques et les notes de la gamme.
3. Puis effectuez des tests similaires pour le timbre et le volume sonore.

Lorsque vous désirez passer à la leçon suivante, tapez **CLOAD** et appuyez sur **RETURN** deux fois. Lorsque le message **READY** apparaît à l'écran, tapez **RUN** et appuyez sur **RETURN** pour commencer le cours.

LEÇON 2 : EFFETS SONORES

§ 6 - Résumé de la leçon 2

1. Il est possible d'utiliser des variables dans l'instruction SOUND.

```
SOUND 0,X,Y,Z
SOUND 0,HAUTEUR,10,8
SOUND 0,15,TIMBRE,INTENSITE
```

2. Utilisez une boucle FOR/NEXT pour améliorer la qualité de vos effets sonores.

```
10 FOR HAUTEUR = 1 TO 255
20 SOUND 0,HAUTEUR,10,8
30 NEXT HAUTEUR
```

3. Vous pouvez bien sûr placer n'importe quel paramètre dans une boucle FOR/NEXT.

```
10 FOR INTENSITE = 15 TO 0 STEP -1
20 SOUND 0,15,0,INTENSITE
30 NEXT INTENSITE
```

4. Des bruits d'explosion, des cliquetis, etc, sont créés par une variation rapide entre une intensité sonore forte et faible.

§ 7 - Application pratique de la leçon 2

Note: Effacez la mémoire de l'ordinateur avant chaque exercice pratique en tapant **NEW** et en appuyant sur **RETURN**.

Essayez ceci:

```
READY
10 REM TONNERRE
20 FOR TONALITE = 1 TO 255
30 SOUND 0,TONALITE,8,15
40 NEXT TONALITE
```

```
READY
10 REM LOCOMOTIVE A VAPEUR
20 FOR INTENSITE = 15 TO 0 STEP -1
30 SOUND 0,15,0,INTENSITE
40 NEXT INTENSITE
50 GOTO 10
```

```
READY
10 REM SIRENES
20 FOR HAUTEUR = 100 TO 10 STEP -1
30 SOUND 0,HAUTEUR,10,8
40 NEXT HAUTEUR
50 FOR HAUTEUR = 10 TO 100
60 SOUND 0,HAUTEUR,10,8
70 NEXT HAUTEUR
80 GOTO 20
```

```
READY
5 REM ORGUE
10 HAUTEUR = INT(RND(1)*255)
20 FOR INTENSITE = 0 TO 15
30 SOUND 0,HAUTEUR,10,INTENSITE
35 NEXT INTENSITE
40 FOR INTENSITE = 15 TO 0 STEP -1
50 SOUND 0,HAUTEUR,10,INTENSITE
55 NEXT INTENSITE
60 GOTO 10
```

```
READY
10 REM PIANO
20 HAUTEUR = INT(RND(0)*255)
30 FOR INTENSITE = 15 TO 0 STEP -1
40 SOUND 0,HAUTEUR,10,INTENSITE
50 NEXT INTENSITE
60 GOTO 20
```

Ajoutez la ligne suivante pour jouer deux notes à la fois:

```
45 SOUND 1,HAUTEUR + 32,10,INTENSITE
```

Lorsque vous désirez passer à la leçon suivante, tapez **CLOAD** et appuyez sur **RETURN** deux fois. Lorsque le message **READY** apparaît sur l'écran, tapez **RUN** et appuyez sur **RETURN** pour commencer le cours.

LEÇON 3 : PROGRAMMES D'EFFETS SONORES

§ 8 - Programme sonore

Un programme générant une succession d'effets sonores est appelé programme sonore par abus de langage. Les paragraphes 9 à 17 ci-après montrent l'évolution d'un tel programme utilisant un seul registre son (la leçon 7 envisage le cas des programmes utilisant plus d'un registre son à la fois).

§ 9 - Lignes numérotées

```
READY
100 SOUND 0,81,10,8
110 SOUND 0,64,10,8
120 SOUND 0,53,10,8
130 SOUND 0,64,10,8
140 SOUND 0,60,10,8
150 SOUND 0,47,10,8
160 SOUND 0,60,10,8
170 END
```

§ 10 - Ajout de boucles de temporisation

Nous ajoutons maintenant pour chaque son une boucle de retardement.

```
READY
100 SOUND 0,81,10,8
105 FOR D=1 TO 300:NEXT D
110 SOUND 0,64,10,8
115 FOR D=1 TO 300:NEXT D
120 SOUND 0,53,10,8
125 FOR D=1 TO 300:NEXT D
130 SOUND 0,64,10,8
135 FOR D=1 TO 300:NEXT D
140 SOUND 0,60,10,8
145 FOR D=1 TO 300:NEXT D
150 SOUND 0,47,10,8
155 FOR D=1 TO 300:NEXT D
160 SOUND 0,60,10,8
165 FOR D=1 TO 300:NEXT D
170 END
```

§ 11 - Boucles de temporisation de durées variables

```
READY
100 SOUND 0,81,10,8
105 FOR D=1 TO 400:NEXT D
110 SOUND 0,64,10,8
115 FOR D=1 TO 200:NEXT D
120 SOUND 0,53,10,8
125 FOR D=1 TO 400:NEXT D
130 SOUND 0,64,10,8
135 FOR D=1 TO 400:NEXT D
140 SOUND 0,60,10,8
145 FOR D=1 TO 300:NEXT D
150 SOUND 0,47,10,8
155 FOR D=1 TO 200:NEXT D
160 SOUND 0,60,10,8
165 FOR D=1 TO 600:NEXT D
170 END
```

§ 12 - Sous-programme de temporisation

```
READY
100 SOUND 0,81,10,8
105 D=400:GOSUB 1000
110 SOUND 0,54,10,8
115 D=200:GOSUB 1000
120 SOUND 0,53,10,8
125 D=400:GOSUB 1000
130 SOUND 0,64,10,8
135 D=400:GOSUB 1000
140 SOUND 0,60,10,8
145 D=300:GOSUB 1000
150 SOUND 0,47,10,8
155 D=200:GOSUB 1000
160 SOUND 0,60,10,8
165 D=600:GOSUB 1000
170 END
1000 FOR T=1 TO D:NEXT T
1010 RETURN
```

§ 13 - Instructions PEEK et POKE

Pour obtenir un tempo très précis dans un programme musical, vous pouvez utiliser l'horloge incorporée dans l'ordinateur. Pour cela, vous devez déjà la remettre à zéro par une instruction POKE. Vous pouvez ensuite lire sa valeur par une instruction PEEK. Le programme ci-dessous remplace avantageusement les boucles FOR/NEXT moins précises.

```
READY
1000 POKE 20,0:REM REMET L'HORLOGE A ZERO
1010 H=PEEK(20):REM LE TEMPS MESURE EST PLACE DANS H
1020 IF H < D THEN GOTO 1010:REM SI LE TEMPS EST TROP COURT, ON RELIT L'HORLOGE
1030 RETURN:REM LE TEMPS EST ECOULE
```

§ 14 - Application pratique de la leçon 3

Note: Effacez la mémoire de l'ordinateur avant chaque exercice pratique en tapant **NEW** et en appuyant sur **RETURN**.

Tapez le programme suivant et écoutez le résultat.

```
READY
100 REM OISEAUX
110 P=INT(RND(0)*6)+250
120 V=INT(RND(0)*4)+6
130 SOUND 0,P,14,V
140 FOR X=1 TO 10:NEXT X
150 GOTO 110
160 END
```

Modifiez maintenant ce petit programme et écoutez la différence !

```
READY
110 P=INT(RND(0)*256)
120 V=INT(RND(0)*16)
140 GOTO 10
```

Effacez la ligne 150 en tapant:

150 et en appuyant sur la touche **RETURN**

N'hésitez pas à modifier chacune des valeurs afin de bien pouvoir juger de l'effet de chaque paramètre.

```
READY
100 REM CHUTE D'UNE PIECE
105 P=60
110 FOR J=10 TO 0 STEP -0.2
115 REM AUGMENTATION DE L'INTENSITE
120 FOR V=1 TO J:SOUND 0,P,10,V:NEXT V
122 REM DIMINUTION DE L'INTENSITE
125 FOR V=2*J TO 1 STEP -1:SOUND 0,P,10,V:NEXT V
130 NEXT J
140 END
```

Lorsque vous désirez passer à la leçon suivante, tapez **CLOAD** et appuyez sur **RETURN** deux fois. Lorsque le message **READY** apparaît à l'écran, tapez **RUN** et appuyez sur **RETURN** pour commencer le cours.

LEÇON 4 : QUELQUES PROGRAMMES PLUS EFFICACES

§ 15 - Programmes de musique comportant un sous-programme de temporisation

```
READY
100 H=81:D=40:GOSUB 1000
110 H=64:D=20:GOSUB 1000
120 H=53:D=40:GOSUB 1000
130 H=64:D=40:GOSUB 1000
140 H=60:D=40:GOSUB 1000
150 H=45:D=20:GOSUB 1000
160 H=60:D=60:GOSUB 1000
170 END
1000 SOUND 0,H,10,8
1010 POKE 20,0
1020 C=PEEK(20)
1030 IF C < D THEN GOTO 1020
1040 RETURN
```

§ 16 - Utilisation des instructions **READ** et **DATA**

```
READY
100 READ H,D
110 IF H=0 AND D=0 THEN END
120 GOSUB 1000
130 GOTO 100
1000 SOUND 0,H,10,8
1010 POKE 20,0
1020 C=PEEK(20)
1030 IF C < D THEN GOTO 1020
1040 RETURN
2000 DATA 81,40
2010 DATA 64,20
2020 DATA 53,40
2030 DATA 64,50
2040 DATA 60,40
2050 DATA 47,20
2060 DATA 60,60
2070 DATA 0,0
```

§ 17 - Gagnons encore en efficacité

```
READY
2000 DATA 81,40,64,20,53,40,64,40,60,40
2010 DATA 47,20,60,60,0,0
```

§ 18 - Applications pratiques de la leçon 4

Note: Effacez la mémoire de l'ordinateur avant chaque exercice en tapant **NEW** et en appuyant sur **RETURN**.

Tapez soigneusement le programme ci-dessous. Les lignes de **DATA** contiennent les notes de la cinquième symphonie de Beethoven.

```
READY
100 READ H,D
110 IF H=0 AND D=0 THEN END
120 GOSUB 1000
130 GOTO 100
1000 SOUND 0,H,10,8
1010 POKE 20,0
1020 C=PEEK(20)
1030 IF C < D THEN GOTO 1020
1040 RETURN
2000 DATA 81,30
2010 DATA 81,30
2020 DATA 81,30
2030 DATA 102,90
2040 DATA 0,0
```

Quelque chose ne va pas? Effectivement, il n'y a pas de pause entre les notes. Voici un moyen simple d'y remédier: ajoutez cette ligne à votre programme:

```
1040 SOUND 0,0,0,0:RETURN
```

Maintenant, chaque note s'arrête avant que la suivante soit jouée.

Voici à présent la liste des valeurs numériques des hauteurs et des durées de notes pour une autre mélodie. Entrez-les dans le programme précédent, à l'intérieur des lignes de **DATA**. Assurez-vous bien que le dernier couple de valeurs est bien 0,0.

H	D
121	30
121	30
60	30
60	30
53	30
53	30
60	60
0	0

Voici quelques autres exemples.

```

READY
2000 DATA 121,10,121,10,121,20,91,10,
91,10,91,20,96,10,91,10,81,10,
72,10,68,10,81,10,72,30
2010 DATA 68,10,60,20,53,10,68,10,
72,10,91,10,81,20,91,60
2020 DATA 0,0
  
```

```

READY
2000 DATA 85,40,85,40,72,40,72,
40,108,40,108,40,144,40,108,20
85,20,72,120
2010 DATA 64,20,85,20,72,120,72
40,85,40,85,40,72,40,72,40,108,
40,108,40,144,40,108,20,96,20
2020 DATA 85,40,72,80,85,40,96,
80,108,80
2030 DATA 0,0
  
```

```

READY
2000 DATA 81,30,81,45,96,15,96,30,
81,30,81,45,108,15,108,30,96,30,
91,30,81,30,72,30,64,30,81,90
2010 DATA 81,30,81,45,96,15,96,30,
81,30,81,45,108,15,108,30,53,30,
57,30,53,30,47,30,72,30,53,90
2020 DATA 81,30,47,45,47,15,53,30,
60,30,60,45,64,15,64,30,60,30,53,
30,64,30,72,30,81,30,60,90
2030 DATA 60,30,60,45,72,15,72,30,
60,30,60,45,81,15,81,30,81,30,72,
30,60,30,81,30,53,30,60,90
2040 DATA 0,0
  
```

Enfin, vous pouvez utiliser le programme suivant pour que les musiques précédentes semblent être interprétées par un piano:

```


READY
10 REM PIANO
100 READ H,D
110 IF H=0 AND D=0 THEN END
120 GOSUB 1000
130 GOTO 100
1000 FOR INTENSITE = 15 TO 0 STEP -0.5
1001 SOUND 0,H,10,INTENSITE
1002 NEXT INTENSITE
1010 POKE 20,0
1020 C=PEEK(20)
1030 IF C < D THEN GOTO 1020
1040 RETURN
  
```

Lorsque vous désirez passer à la leçon suivante, tapez **CLOAD** et appuyez sur **RETURN** deux fois. Lorsque le message **READY** apparaît à l'écran, tapez **RUN** et appuyez sur **RETURN** pour commencer le cours.

LEÇON 5 : GAMMES MUSICALES

§ 19 - Portées

En musique, les notes représentées par des symboles déterminent la hauteur de chaque son, c'est-à-dire sa fréquence. Les notes sont placées sur une portée qui consiste en cinq lignes séparées par quatre espaces égaux. Chaque ligne et chaque espace représente une hauteur différente. Les deux portées généralement utilisées correspondent aux clés de sol et clé de fa. La clé de sol correspond à la partie droite d'un clavier de piano, tandis que la clé de fa correspond à la partie gauche. Un signe particulier apparaît au début de chaque portée pour indiquer la clé utilisée. Trois notes, le Si, le Do et le Ré médium, sont situées entre les deux portées.



§ 20 - Informations complémentaires

Les notes sont regroupées en mesures, une mesure étant composée de temps. A l'intérieur d'une partition, toutes les mesures ont la même longueur; l'unité de temps est fixe.

On indique le nombre de temps dans une mesure par un chiffre placé à droite de la clé, en haut.


On indique l'unité de temps par un chiffre placé à droite de la clé, en bas. Notez qu'il n'est pas vraiment utile de posséder à fond ces différentes notions pour pouvoir transcrire des mélodies simples d'une partition sur un ordinateur ATARI.

Lorsque vous désirez passer à la leçon suivante, tapez **CLOAD** et appuyez sur **RETURN** deux fois. Lorsque le message **READY** apparaît sur l'écran, tapez **RUN** et appuyez sur **RETURN** pour commencer le cours.


LEÇON 6 : IDENTIFIONS LES NOTES

§ 21 - Lignes supplémentaires

Les lignes supplémentaires sont des petites lignes utilisées pour les notes très aiguës (c'est-à-dire situées au-dessus de la portée) ou très graves (c'est-à-dire situées en-dessous). Voici quelques exemples concernant la clé de Sol.



Voici quelques exemples concernant la clé de Fa.



§ 22 - Correspondances entre notes et valeurs numériques

Notes aiguës	DO	29
	SI	31
	LA DIESE/SI BEMOL	33
	LA	35
	SOL DIESE/LA BEMOL	37
	SOL	40
	FA DIESE/SOL BEMOL	42
	FA	45
	MI	47
	RE DIESE/MI BEMOL	50
	RE	53
	DO DIESE/RE BEMOL	57
	DO	60
	SI	64
	LA DIESE/SI BEMOL	68
	LA	72
	SOL DIESE/LA BEMOL	76
	SOL	81
	FA DIESE/SOL BEMOL	85
FA	91	
MI	96	
RE DIESE/MI BEMOL	102	
RE	108	
DO DIESE/RE BEMOL	114	
Medium	DO	121
	SI	128
	LA DIESE/SI BEMOL	136
	LA	144
	SOL DIESE/LA BEMOL	153
	SOL	162
	FA DIESE/SOL BEMOL	173
Notes graves	FA	182
	MI	193
	RE DIESE/MI BEMOL	204
	RE	217
	DO DIESE/RE BEMOL	230
	DO	243

§ 23 - Applications pratiques de la leçon 6

Note: Effacez la mémoire de l'ordinateur avant chaque exercice en tapant **NEW** et en appuyant sur **RETURN**.

Voici à nouveau le programme de musique que nous avons déjà tapé dans les exercices précédents. Utilisez-le à nouveau pour jouer les partitions de cette page.

```

100 READ H,D
110 IF H=0 AND D=0 THEN END
120 GOSUB 1000
130 GOTO 100
1000 SOUND 0,H,10,8
1010 POKE 20,0
1020 C=PEEK(20)
1030 IF C < D THEN GOTO 1020
1040 SOUND 0,0,0,0: RETURN
2000 DATA ...

```

Pour chaque note d'une partition ci-dessous, vous devez placer dans une ligne de DATA deux valeurs numériques: la hauteur en premier (voir tableau précédent et § 20), et la durée (utilisez pour cela la correspondance ci-dessous pour la blanche, la noire, la croche, la double croche).

♩ = 120 ♪ = 60 ♫ = 30 ♬ = 15



Lorsque vous désirez passer à la leçon suivante, tapez **CLOAD** et appuyez sur **RETURN** deux fois. Lorsque le message **READY** apparaît sur l'écran, tapez **RUN** et appuyez sur **RETURN** pour commencer la leçon.

LEÇON 7 : EFFETS SONORES A PLUSIEURS VOIX

§ 24 - Programme de musique jouant une seule note à la fois

```

100 READ H,D
110 IF H=0 AND D=0 THEN END
120 GOSUB 1000
130 GOTO 100
1000 SOUND 0,H,10,8
1010 POKE 20,0
1020 C=PEEK(20)
1030 IF C < D THEN GOTO 1020
1040 SOUND 0,0,0,0: RETURN
2000 DATA 81,40
2010 DATA 64,20
2020 DATA 53,40
2030 DATA 64,50
2040 DATA 60,40
2050 DATA 47,20
2060 DATA 60,60
2070 DATA 0,0

```

§ 25 - Résumé de la leçon 7

Vous pouvez produire toute une variété d'effets sonores en utilisant 2 ou plusieurs registres sons. Toutefois, pour que l'emploi de plusieurs registres sonores ne produise pas un ensemble discordant, vous devez choisir des notes de manière à ce qu'elles soient harmonieuses entre elles. Pour cela, vous devez déjà choisir un timbre pur pour chaque registre (valeur 10).

Les notes de musique tendent à devenir dissonantes lorsque l'écart qui les sépare se réduit. Un chœur ou un accord est un ensemble de plusieurs voix jouées simultanément.

§ 26 - Applications pratiques de la leçon 7

Note: Effacez la mémoire de l'ordinateur avant chaque exercice en tapant **NEW** et en appuyant sur **RETURN**.

Essayez le programme suivant. Il s'agit de deux effets sonores que nous combinons pour obtenir un effet intéressant.

```
READY
5 REM SIFFLET DE TRAIN
10 FOR SIRENE=1 TO 2
15 SOUND 0,190,10,8:SOUND 1,175,10,8
20 FOR DELAI=1 TO 750:NEXT DELAI
30 SOUND 0,0,0,0:SOUND 1,0,0,0
40 FOR DELAI=1 TO 50:NEXT DELAI
50 NEXT SIRENE
```

Voici un autre effet.

```
READY
10 FOR HAUTEUR=100 TO 200
20 SOUND 0,HAUTEUR,10,8
30 SOUND 1,300-HAUTEUR,10,8
40 FOR DELAI=1 TO 5:NEXT DELAI
50 NEXT HAUTEUR
60 GOTO 10
```

Voici un programme musical. Tapez-le soigneusement et écoutez le résultat. Notez que lorsque les quatre registres sons sont utilisés simultanément, il ne faut pas dépasser 8 en intensité maximale pour chaque voix.

```
100 READ H0,H1,H2,H3,D
110 IF H0=0 AND H1=0 AND H2=0 AND H3=0 AND D=0 THEN END
120 GOSUB 1000
130 GOTO 100
1000 SOUND 0,H0,10,8:SOUND 1,H1,10,8:SOUND 2,H2,10,8:SOUND 3,H3,10,8
1010 POKE 20,0
1020 C=PEEK(20)
1030 IF C < D THEN GOTO 1020
1040 SOUND 0,0,0,0:SOUND 1,0,0,0:SOUND 2,0,0,0:SOUND 3,0,0,0:RETURN
2000 DATA 121,96,81,60,20
2010 DATA 162,121,96,81,20
2020 DATA 172,121,96,72,20
2030 DATA 162,121,96,81,40
2040 DATA 108,92,82,64,20
2050 DATA 162,121,96,60,100
2060 DATA 0,0,0,0,0
```

Tapez maintenant les lignes de DATA suivantes:

```
READY
2000 DATA 243,193,162,60,20
2010 DATA 243,193,162,60,20
2020 DATA 243,193,162,60,20
2030 DATA 243,193,162,61,20
2040 DATA 182,144,121,72,20
2050 DATA 182,144,121,72,20
2060 DATA 243,193,162,81,40
2070 DATA 0,0,0,47,20
2080 DATA 0,0,0,47,20
2090 DATA 0,128,108,53,20
2100 DATA 0,128,108,53,20
2110 DATA 243,193,162,60,80
3000 DATA 0,0,0,0,0
```

Le programme ci-dessous est similaire au précédent mais les effets sonores ressemblent davantage à ceux d'un piano. Vous pouvez gagner du temps en ré-utilisant la plupart des lignes du programme précédent et en changeant seulement les lignes qui diffèrent.

```
10 REM PIANO
100 READ H0,H1,H2,H3,D
110 IF H0=0 AND H1=0 AND H2=0 AND H3=0 AND D=0 THEN END
120 GOSUB 1000
130 GOTO 100
1000 FOR INTENSITE=8 TO 0 STEP -0.5
1002 SOUND 0,H0,10,INTENSITE:SOUND 1,H1,10,INTENSITE:SOUND 2,H2,10,INTENSITE:
SOUND 3,H3,10,INTENSITE
1005 NEXT INTENSITE
1010 POKE 20,0
1020 C=PEEK(20)
1030 IF C < D THEN GOTO 1020
1040 RETURN
2000 DATA 243,193,162,60,0
2010 DATA 243,193,162,60,0
2020 DATA 243,193,162,60,0
2030 DATA 243,193,162,81,0
2040 DATA 182,144,121,72,0
2050 DATA 182,144,121,72,0
2060 DATA 243,193,162,81,20
2070 DATA 0,0,0,47,0
2080 DATA 0,0,0,47,0
2090 DATA 0,128,108,53,0
2100 DATA 0,128,108,53,0
2110 DATA 243,193,162,60,16
3000 DATA 0,0,0,0,0
```

```

READY
10 REM ORGUE
100 READ H0,H1,H2,H3,D
110 IF H0=0 AND H1=0 AND H2=0 AND H3=0 AND D=0 THEN END
120 GOSUB 1000
130 GOTO 100
1000 FOR INTENSITE=0 TO 8
1002 SOUND 0,H0,10,INTENSITE:SOUND 1,H1,10,INTENSITE:SOUND 2,H2,10,INTENSITE:
SOUND 3,H3,10,INTENSITE
1005 NEXT INTENSITE
1010 POKE 20,0
1020 C=PEEK(20)
1030 IF C < D THEN GOTO 1020
1036 FOR INTENSITE=8 TO 0 STEP -0.5
1037 SOUND 0,H0,10,INTENSITE:SOUND 1,H1,10,INTENSITE:SOUND 2,H2,10,INTENSITE:
SOUND 3,H3,10,INTENSITE
1038 NEXT INTENSITE
1040 RETURN
2000 DATA 243,193,162,60,20
2010 DATA 243,193,162,60,20
2020 DATA 243,193,162,60,20
2030 DATA 243,193,162,81,20
2040 DATA 182,144,121,72,20
2050 DATA 182,144,121,72,20
2060 DATA 243,193,162,81,40
2070 DATA 0,0,0,47,20
2080 DATA 0,0,0,47,20
2090 DATA 0,128,108,53,20
2100 DATA 0,128,108,53,20
2110 DATA 243,193,162,60,80
3000 DATA 0,0,0,0,0
    
```

Vous êtes arrivé au questionnaire final de cette cassette. La piste sonore n'est pas utilisée sur cette dernière partie.

§ 27 - Programme utilisé par le questionnaire final

```

READY
100 READ H0,H1,H2,H3,D
110 IF H0=0 AND H1=0 AND H2=0 AND H3=0 AND D=0 THEN END
120 GOSUB 1000
130 GOTO 100
1000 SOUND 0,H0,10,8:SOUND 1,H1,10,8:SOUND 2,H2,10,8:SOUND 3,H3,10,8
1010 POKE 20,0
1020 H=PEEK(20)
1030 IF H < D THEN GOTO 1020
1040 SOUND 0,0,0,0:SOUND 1,0,0,0:SOUND 2,0,0,0:SOUND 3,0,0,0:RETURN
2000 DATA 121,96,81,60,20
2010 DATA 162,121,96,81,20
2020 DATA 172,121,96,72,20
2030 DATA 162,121,96,81,40
2040 DATA 108,92,82,64,20
2050 DATA 162,121,96,60,100
2060 DATA 0,0,0,0,0
    
```

SOMMAIRE DES COURS DE LA CASSETTE B INTRODUCTION AUX GRAPHIQUES

- Leçon 1 : DEMONSTRATIONS GRAPHIQUES
Affichage de quelques effets simples obtenus en mode graphique.
- Leçon 2 : CHANGER DE COULEUR
Instruction permettant le changement de couleur et description de l'utilisation des registres couleurs.
- Leçon 3 : MODES GRAPHIQUES 1 ET 2
Introduction aux modes graphiques et description détaillée des registres couleurs.
- Leçon 4 : DE VRAIS GRAPHIQUES
Utilisation des instructions PLOT et DRAWTO dans les différents modes graphiques.
- Leçon 5 : QUELQUES ASTUCES
Comment utiliser l'écran tout entier en mode graphique et passer d'un mode à un autre sans effacer la mémoire d'écran.
- Leçon 6 : POSITION ET ANIMATION
Comment positionner le texte et donner l'apparence des mouvements.
- Leçon 7 : CARACTERES GRAPHIQUES
Comment dessiner avec des caractères spéciaux.

QUESTIONNAIRE FINAL : Graphiques. Révision des différents points vus dans ce cours.

Cassette B	Leçons	Valeurs au compteur du magnétocassette		
		Début du chargement	Début de la voix	Fin de la voix
Face 1	1	000	33	56
	2	56	99	156
	3	156	184	220
	4	290	309	368
Face 2	5	000	39	108
	6	108	140	221
	7	221	249	288
	Test final	288		

Figure 3 - Valeurs du compteur du magnétocassette pour les programmes de la cassette B

DESCRIPTION DES DIFFERENTES PARTIES DE LA CASSETTE B

LEÇON 1 : DEMONSTRATIONS GRAPHIQUES

Lorsque vous désirez passer à la leçon suivante, tapez **CLOAD** et appuyez sur **RETURN** deux fois. Lorsque le message **READY** apparaît à l'écran, tapez **RUN** et appuyez sur **RETURN** pour commencer le cours.

LEÇON 2 : CHANGER DE COULEUR

§ 1 - Table des couleurs des ordinateurs ATARI

Reportez-vous à la table ci-dessous pour trouver l'équivalence entre une couleur et le numéro de code utilisé par les ordinateurs ATARI.

ATTENTION: LES VALEURS AINSI QUE TOUTES LES INFORMATIONS SE RAPPORTANT AUX COULEURS ET AUX LUMINANCES (BRILLANCE) DANS CE COURS S'APPLIQUENT SEULEMENT AUX ORDINATEURS FONCTIONNANT SUR DES TELEVISEURS RECEVANT LES EMISSIONS NTSC ET PAL, C'EST-A-DIRE AUX ORDINATEURS VENDUS DANS TOUS LES PAYS SAUF LA FRANCE. EN EFFET, SUR LES ORDINATEURS FRANÇAIS, L'INSTALLATION D'UNE PRISE PERITELEVISION MODIFIE LES POSSIBILITES COULEURS. L'INFORMATION COULEUR N'EST PLUS UTILISEE. PAR CONTRE, L'INFORMATION LUMINANCE DEVIENT UNE INFORMATION COULEUR. EN OUTRE, SEULES 8 COULEURS SONT DISPONIBLES. PAR CONTRE, LA QUALITE GRAPHIQUE EST EXCELLENTE.

VEUILLEZ NOTER QUE LES ORDINATEURS VENDUS EN FRANCE PEUVENT TOUTEFOIS TOUJOURS FONCTIONNER EN PAL PAR LE CÂBLE D'ANTENNE ORDINAIRE. SI DONC VOUS POSSEDEZ UN TELEVISEUR MULTISTANDARD, CAPABLE DE RECEVOIR LES EMISSIONS CODEES EN PAL, VOUS RETROUVerez TOUTES LES POSSIBILITES D'ORIGINE DE LA MACHINE.

LA TABLE CI-DESSOUS N'EST VALABLE QUE POUR LES ORDINATEURS PAL ET NTSC.

Numéro de couleur Couleur approximative pour une luminance moyenne

0	gris
1	or
2	orange
3	rouge orangé
4	rose
5	pourpre ou violet
6	violet
7	bleu
8	bleu foncé
9	bleu clair
10	turquoise
11	vert bleu
12	vert
13	jaune vert
14	orange vert
15	orange clair

§ 2 - Registres couleurs

L'ordinateur ATARI comporte 5 registres couleurs numérotés de 0 à 4. Chaque registre couleur correspond à une application spécifique. Ainsi par exemple, le registre couleur 4 change la couleur de la bordure de l'écran tandis que le registre couleur 2 change la couleur de fond de l'écran lui-même. Chaque registre reçoit deux valeurs: l'une correspond à la couleur, l'autre à la luminance ou brillance.

§ 3 - Instruction SETCOLOR

L'utilisation de l'instruction SETCOLOR permet d'assigner une couleur spécifique et sa luminance à un registre couleur donné. La syntaxe de cette instruction est la suivante:

```
SETCOLOR X,Y,Z
```

Où X représente le numéro du registre couleur
Y le numéro de la couleur
Z le numéro de la luminance.

§ 4 - Information complémentaire sur l'instruction SETCOLOR

La luminance est modifiée par les valeurs paires de Z (0, 2, 4, 6, 8, 10, 12, et 14).

Note: Si vous utilisez une luminance de 10 pour le registre couleur 2, les caractères seront de la même nuance que le fond de l'écran. En conséquence, vous ne les distinguerez pas. Pour corriger cette situation, donnez une nouvelle valeur de luminance ou appuyez sur **SYSTEM RESET**. Attention ! Cette remarque ne s'applique pas aux ordinateurs vendus en France. Pour eux, une modification de luminance provoque un changement de couleur. Des variables peuvent être utilisées comme arguments de l'instruction SETCOLOR. Un exemple en est donné ci-dessous.

§ 5 - Application pratique de la leçon 2

Note: Effacez la mémoire de l'ordinateur avant chaque exercice pratique. Pour cela, tapez **NEW** et appuyez sur **RETURN**.

N'hésitez pas à utiliser l'instruction SETCOLOR avec les registres 2 ou 4. Utilisez cette commande soit en mode direct, soit en tapant le programme ci-dessous permettant de visualiser toutes les couleurs et toutes les nuances disponibles sur votre ordinateur ATARI.

```
READY
10 REM PROGRAMME DE DEMONSTRATION DES
COULEURS ET DES NUANCES
20 REM UTILISANT LE REGISTRE 2
30 FOR X=0 TO 15
40 FOR Y=0 TO 14
50 SETCOLOR 2,X,Y
55 FOR I=0 TO 100:NEXT I
60 NEXT Y
70 NEXT X
80 END
```

Etudiez également l'instruction SETCOLOR avec le registre 1 pour examiner les différences de nuance sur les lettres de l'écran. Si les lettres disparaissent lors d'un de vos essais, c'est que vous venez de leur attribuer la même couleur que celle du fond. Tapez une autre instruction SETCOLOR ou appuyez sur la touche **SYSTEM RESET**.

```
READY
10 REM ROCKET
20 SETCOLOR 2,0,2
25 PRINT CHR$(125):REM EFFACE L'ECRAN
30 FOR COMPTEUR = 10 TO 0 STEP -1
40 PRINT COMPTEUR
50 FOR TEMPS = 1 TO 100:NEXT TEMPS
60 NEXT COMPTEUR
65 SETCOLOR 2,4,14
70 PRINT CHR$(253);"BOUM!"
80 SETCOLOR 2,0,2
90 END
```

Lorsque vous désirez passer à la leçon suivante, tapez **CLOAD** et appuyez sur **RETURN** deux fois. Lorsque le message **READY** apparaît à l'écran, tapez **RUN** et appuyez sur **RETURN** pour commencer la leçon.


LEÇON 3 : MODES GRAPHIQUES 1 ET 2

§ 6 - Modes graphiques

INSTRUCTION	ASPECT DES OBJETS SUR L'ECRAN	TAILLE
GRAPHICS 0	Mode standard	24 lignes de 40 caractères
GRAPHICS 1	Double largeur	20 lignes de 20 caractères
GRAPHICS 2	Double largeur et double hauteur	10 lignes de 20 caractères
GRAPHICS 3	Grands carrés graphiques	20 lignes de 40 carrés
GRAPHICS 4	Carrés plus petits	40 lignes de 80 carrés
GRAPHICS 5	Carrés plus petits (plus de couleur)	40 lignes de 80 carrés

§ 7 - Couleurs dans le mode graphique 1


Les registres couleur 2 et 4 sont utilisés dans le mode graphique 1. Le registre couleur 2 donne la couleur de fond de la partie texte apparaissant dans le bas de l'écran. Le registre couleur 4 donne la couleur de la fenêtre graphique. Utilisez l'instruction SETCOLOR pour modifier aussi bien la couleur de la fenêtre texte que celle de la fenêtre graphique.



§ 8 - Affichage sur l'écran en mode graphique 1 ou 2

L'instruction PRINT habituelle permet la mise en place de texte à l'intérieur de la fenêtre texte. Pour pouvoir afficher des caractères dans la partie graphique de l'écran, il faut utiliser l'instruction PRINT avec une syntaxe un peu particulière:

```
PRINT # 6;"TEXTE"
```



§ 9 - Comment obtenir des textes de plusieurs couleurs

Selon la manière dont on tape le texte devant apparaître dans la fenêtre graphique c'est-à-dire selon qu'il s'agit de majuscules, de minuscules, ou de lettres en inversion vidéo, la couleur attribuée à chaque lettre peut être changée. Les registres couleur 0, 1, 2 et 3 sont alors utilisés selon le tableau ci-dessous:

REGISTRE COULEUR	TYPE DE CARACTERE	EXEMPLE D'INSTRUCTION PERMETTANT DE CHANGER CES COULEURS
0	Lettres majuscules	SETCOLOR 0,9,10
1	Lettres minuscules	SETCOLOR 1,15,0
2	Lettres majuscules en inversion vidéo	SETCOLOR 2,8,6
3	Lettres minuscules en inversion vidéo	SETCOLOR 3,5,14
4	Couleur de fond et de bordure	SETCOLOR 4,2,2

§ 10 - Couleurs par défaut

Si aucune instruction SETCOLOR n'est donnée avant qu'une instruction PRINT # 6 ne soit exécutée à l'intérieur d'un programme, les couleurs de caractères seront celles choisies par l'ordinateur lors de sa mise sous tension.

REGISTRE COULEUR	TYPE DE CARACTERE	COULEUR PAR DEFAUT
0	Lettres majuscules	Orange
1	Lettres minuscules	Vert clair
2	Lettres majuscules en inversion vidéo	Bleu foncé
3	Lettres minuscules en inversion vidéo	Rouge
4	Fond de l'écran	Noir

§ 11 - Application pratique de la leçon 3

Note: Effacez la mémoire de l'ordinateur avant chaque exercice en tapant NEW et en appuyant sur RETURN.

Passer en mode graphique 1 en tapant GRAPHICS 1 puis en appuyant sur RETURN. Envoyez plusieurs messages vers l'écran en utilisant l'instruction PRINT # 6; réalisez les mêmes opérations en mode graphique 2.

Note: Vous obtenez des caractères en inversion vidéo en appuyant d'abord sur la touche marquée du logo ATARI dans le coin inférieur droit du clavier puis en tapant ensuite les caractères désirés. Pour revenir en mode normal, il suffit d'appuyer une nouvelle fois sur la touche logo ATARI. Pour revenir en mode texte normal, tapez GRAPHICS 0 et appuyez sur RETURN.

§ 12 - Application de l'instruction SETCOLOR aux modes graphiques 1 et 2

Tapez puis essayez ce programme:

```
10 GRAPHICS 2
20 PRINT #6;"UNdeuxTROISquatre"
100 END
```

Remarquez la couleur des lettres dans la fenêtre graphique. Vérifiez ces couleurs avec celles données par le tableau des valeurs par défaut. Maintenant, tapez la suite de l'instruction SETCOLOR ci-dessous en mode direct et tout en restant en mode graphique 1.

```
SETCOLOR 0,8,0      Observez la modification de la couleur de "UN"
SETCOLOR 1,2,10    Observez la modification de la couleur de "deux"
SETCOLOR 2,15,14   Observez la modification de la couleur de "TROIS"
SETCOLOR 3,0,6     Observez la modification de la couleur de "quatre"
SETCOLOR 4,4,2     Remarquez la couleur de fond
```

Listez le programme précédent et ajoutez les lignes d'instructions suivantes:

```
30 FOR X=0 TO 3: REM NUMERO DU REGISTRE COULEUR
40 FOR Y=1 TO 10: REM NOMBRE DE FLASH
50 SETCOLOR X,0,0
60 FOR TEMPS=1 TO 100:NEXT TEMPS: REM TEMPORISATION
70 SETCOLOR X,0,14
80 FOR TEMPS=1 TO 100:NEXT TEMPS: REM TEMPORISATION
90 NEXT Y
95 NEXT X
```

Faites tourner ce programme puis appuyez sur la touche SYSTEM RESET. Ajoutez alors l'instruction suivante:

```
15 SETCOLOR 4,4,10
```

et essayez-le à nouveau.


Note: Chaque fois qu'une instruction GRAPHICS 0, 1, ou 2 est exécutée, l'écran est effacé et les couleurs par défaut sont remises en place dans les registres couleur.

L'abréviation de la commande GRAPHICS est GR. ainsi GRAPHICS 1 peut être abrégé en GR.1.

Lorsque vous désirez passer à la leçon 4, tapez **CLOAD** et appuyez sur **RETURN** deux fois. Lorsque le message **READY** apparaît à l'écran, tapez **RUN** et appuyez sur **RETURN** pour commencer la leçon.


LEÇON 4 : DE VRAIS GRAPHIQUES

§ 13 - Présentation de l'écran en mode graphique 3



§ 14 - Commandes PLOT et DRAWTO

La commande PLOT permet de faire apparaître un carré sur l'écran dans la case de coordonnées X,Y. La valeur de X varie de 0 à 39, la valeur de Y de 0 à 19. Notez que le point de coordonnées 0,0 se situe dans le coin supérieur gauche de l'écran.



§ 15 - Petit exemple

Souvenez-vous que l'instruction COLOR ne place pas une couleur dans l'un des registres. L'instruction COLOR dit simplement à l'ordinateur quel registre il doit utiliser pour colorer le point graphique.

Trouvez l'instruction manquante dans le programme ci-dessous.

```
10 GRAPHICS 3
20 SETCOLOR 3,9,4
30 PLOT 15,6
```

§ 16 - Commande COLOR

Une commande COLOR choisit le registre couleur qui sera utilisé par les instructions PLOT et DRAWTO dans la suite du programme. La table ci-dessous illustre la relation existante entre les instructions COLOR et les registres couleur sélectionnés.

COLOR 1	Correspond au registre couleur 0
COLOR 2	Correspond au registre couleur 1
COLOR 3	Correspond au registre couleur 2
COLOR 4	Correspond au registre couleur 3

§ 17 - Petit exemple utilisant l'instruction COLOR

```
10 GRAPHICS 3
20 SETCOLOR 0,6,8
30 SETCOLOR 2,0,6
40 COLOR 1
50 PLOT 15,15
```

Rappelez-vous que le numéro de la couleur 0 équivaut à du gris et le 6 à du bleu.

§ 18 - Application pratique de la leçon 4

Note: Effacez la mémoire de l'ordinateur avant chaque exercice en tapant **NEW** et en appuyant sur **RETURN**.

Exécutez le programme suivant:

```
10 GRAPHICS 3
20 SETCOLOR 0,4,10
30 COLOR 1
40 PLOT 10,10
```

Changez les instructions COLOR et SETCOLOR pour obtenir des carrés de couleur différente sur l'écran. Puis effacez la mémoire et entrez le programme suivant:

```
10 GRAPHICS 3
20 COLOR 1
30 X=INT(RND(1)*40):Y=INT(RND(1)*20)
40 PLOT X,Y
50 GOTO 20
```

Pour arrêter cet affichage aléatoire, appuyez sur la touche **BREAK**. Ajoutez les lignes suivantes au programme:

```
20 C=INT(RND(1)*4)
25 COLOR C
```

§ 19 - Application pratique de la leçon 4 en utilisant les instructions PLOT et DRAWTO

Essayez ce programme:

```
10 REM ECLAIRAGE CLIGNOTANT
20 GRAPHICS 3
30 COLOR 1
40 PLOT 10,0
42 DRAWTO 20,19
```

Ajoutez les lignes suivantes:

```
45 FOR X=1 TO 10
50 SETCOLOR 0,X,8
60 FOR D=1 TO 100:NEXT D:REM ATTENTE
70 SETCOLOR 0,9,14
80 FOR D=1 TO 100:NEXT D
100 NEXT X
```

Lorsque vous désirez passer à la leçon 5, tapez **CLOAD** et appuyez sur **RETURN** deux fois. Lorsque le message **READY** apparaît sur l'écran, tapez **RUN** et appuyez sur **RETURN** pour commencer la leçon. N'oubliez pas de retourner la cassette et de remettre le compteur du magnétocassette à 0.

LEÇON 5 : QUELQUES ASTUCES

§ 20 - Aspect de l'écran dans les modes graphiques 3, 4 et 5


MODE GRAPHIQUE

ECRAN

REGISTRE
COULEUR

TAILLE
MEMOIRE

3




Ecran graphique:
registre 0, 1 et 2

273 octets

Couleur de fond:
registre 4

4




Ecran graphique:
registre 0

537 octets

Couleur de fond:
registre 4

5



Ecran graphique:
registres 0, 1 et 2


1017 octets

Couleur de fond:
registre 4

§ 21 - Graphics n + 16

Nous avons vu jusqu'à présent que lorsqu'on passait en mode graphique, une fenêtre texte apparaissait dans le bas de l'écran, laissant ainsi quatre lignes en mode graphique GR 0. Si l'on veut supprimer ces lignes et passer tout l'écran en mode graphique, il suffit d'ajouter +16 après le numéro du mode graphique.

GRAPHICS 3




GRAPHICS 3+16




§ 22 - Changement de mode sans effacer l'écran

GRAPHICS 1



GRAPHICS 2+32



§ 23 - Application pratique de la leçon 5

Note: Effacez la mémoire de l'ordinateur avant chaque exercice en tapant NEW et en appuyant sur RETURN.

Tappez ce programme:

```
10 REM ECRIRE SUR TOUT L'ECRAN
20 GRAPHICS 2+16
30 FOR X=1 TO 12
40 PRINT #6;"LIGNE ";X
50 NEXT X
80 FOR Y=1 TO 1000:NEXT Y:REM ATTENTE
90 END
```

Ajoutez maintenant ces deux lignes:

```
60 FOR Y=1 TO 1000:NEXT Y:REM ATTENTE
70 GRAPHICS 1+16+32
```

Note: L'ordinateur utilise dans ce programme différents modes graphiques. En conséquence, la taille de la mémoire varie. Si vous voyez apparaître dans le haut de l'écran quelques caractères inhabituels et parasites, ne vous inquiétez pas, ceci est simplement dû au changement de taille de la mémoire écran.

Lorsque vous désirez passer à la leçon 6, tapez **CLOAD** et appuyez sur **RETURN** deux fois. Lorsque le message **READY** apparaît sur l'écran, tapez **RUN** et appuyez sur **RETURN** pour commencer la leçon.

LEÇON 6 : POSITIONS ET ANIMATIONS

§ 24 - Programme simple d'application utilisant POSITION

La commande **PLOT** place un carré de couleur à un point déterminé par les coordonnées X et Y sur l'écran. L'instruction **POSITION** permet une action similaire en précisant par les coordonnées X et Y le point visé. **POSITION** ne dessine pas le point lui-même. **POSITION** peut être utilisé en mode texte.

```
10 GRAPHICS 0
20 SETCOLOR 2,14,10
30 POSITION 12,11
40 PRINT "MILIEU"
```

Note: L'instruction **POSITION** indique seulement à l'ordinateur où sera affiché le prochain caractère. Une instruction **PRINT**, **PRINT #6**, ou **DRAWTO** est nécessaire après une instruction **POSITION** pour réellement faire apparaître le point à l'écran.

§ 25 - Instruction POSITION

La syntaxe de cette instruction est:

```
POSITION X,Y      où X représente la coordonnée selon un axe horizontal et
                  Y la coordonnée selon un axe vertical.
```

§ 26 - Trouvez le programme exact

Programme 1

```
10 PRINT "TEXTE"
20 POSITION 5,5
30 GRAPHICS 0
```

Programme 2

```
10 GRAPHICS 2
20 POSITION 8,4
30 PRINT #6;"TEXTE"
```

Programme 3

```
10 GRAPHICS 1
20 POSITION "TEXTE"
```

§ 27 - Exemple d'animation

```
READY
10 GRAPHICS 1
20 FOR X=1 TO 10
30 POSITION X,5
40 PRINT #6;"MOT"
50 POSITION X,5
60 PRINT #6;" "
70 NEXT X
```

§ 28 - Autre exemple d'animation

```
READY
10 GRAPHICS 1
20 FOR X=1 TO 10
30 POSITION X,5
40 PRINT #6;"MOT"
45 FOR TEMPS=1 TO 5:NEXT TEMPS
50 POSITION X,5
60 PRINT #6;" "
70 NEXT X
```

§ 29 - Principe d'un texte animé

1. Positionner et afficher le mot à l'écran.
2. Temporiser.
3. Effacer le mot.
4. Afficher le mot à une nouvelle position.

§ 30 - Application pratique de la leçon 6

Note: Effacez la mémoire de l'ordinateur avant chaque exercice en tapant **NEW** et en appuyant sur **RETURN**.

Utilisation de l'instruction **POSITION** en **GRAPHICS 0**:

```
READY
10 GRAPHICS 0
20 PRINT "Tapez deux nombres --le premier est ";PRINT "la valeur X, le second la valeur Y: "
30 PRINT:PRINT
40 INPUT X,Y
50 POSITION X,Y
60 PRINT "POSITION ";X;" ";Y
70 END
```

Positionnement aléatoire:

```
READY
10 REM POSITIONNEMENT ALEATOIRE
20 GRAPHICS 2
30 POSITION 5,5
40 PRINT #6;"BONBON"
50 X=INT(RND(1)*20):Y=INT(RND(1)*10)
60 POSITION X,Y:PRINT #6;"O"
70 GOTO 30
```

Mouvement d'une flèche à travers l'écran:

```
READY
10 GRAPHICS 2
20 FOR X=1 TO 18
30 POSITION X,5
40 PRINT #6;"- "
50 FOR D=1 TO 100:NEXT D
55 POSITION X,5
60 PRINT #6;" "
70 NEXT X
```

Dans l'exemple ci-dessous, nous n'utilisons plus d'instruction PRINT pour effacer le dessin avant de la déplacer. Tapez ce programme et essayez de comprendre la raison de son bon fonctionnement.

```
READY
10 GRAPHICS 2
20 FOR X=1 TO 17
30 POSITION X,5
40 PRINT #6;"- "
50 FOR D=1 TO 100:NEXT D
70 NEXT X
```

En plaçant un caractère blanc (un espace) devant la flèche à l'intérieur des guillemets, vous créez l'illusion du mouvement chaque fois qu'une nouvelle position est donnée sur l'écran. Cela marche très bien tant que la flèche se déplace sur une horizontale.

Lorsque vous désirez passer à la leçon 7, tapez CLOAD et appuyez sur RETURN deux fois. Lorsque le message READY apparaît sur l'écran, tapez RUN et appuyez sur RETURN pour commencer la leçon.

LEÇON 7 : CARACTERES GRAPHIQUES

§ 31 - Instruction POKE




Figure 4 - Touches des caractères de contrôle

(Sur les ordinateurs ATARI 400 ou ATARI 800 les touches SHIFT et LOGO ATARI sont inversées)

Pour utiliser les minuscules et les caractères graphiques sur l'écran dans les modes graphiques 1 et 2, utilisez une instruction POKE.

1. Spécifiez le mode graphique désiré
2. Tapez POKE 756,226
3. Tapez PRINT #6;"lettres minuscules"

§ 32 - Exemple de programme provoquant l'affichage de coeurs

```
READY
10 GRAPHICS 1
20 POKE 756,226
30 PRINT #6;"GRAPHIQUES
```

§ 33 - Exemple de démonstration avec un fond net

```
READY
10 GRAPHICS 1
20 POKE 756,226
30 SETCOLOR 0,0,0
40 PRINT #6;"GRAPHIQUES "
```

§ 34 - Application pratique de la leçon 7

Note: Effacez la mémoire de l'ordinateur avant chaque exercice en tapant NEW et en appuyant sur RETURN.

Entrez le programme suivant. Les caractères graphiques utilisés sont les suivants:

ligne supérieure : CTRL + M
ligne inférieure : CTRL + N
ligne de gauche : CTRL + V
ligne de droite : CTRL + B

```
READY
10 GRAPHICS 2
20 POKE 756,226
30 SETCOLOR 0,0,0
40 PRINT #6;" "
50 PRINT #6;" lettres "
52 PRINT #6;" minuscules "
60 PRINT #6;" "
```

Supprimez la ligne 20 du programme ci-dessus. Remplacez la ligne 20 et supprimez maintenant la ligne 30.

Démonstration d'un programme d'animation:

```
READY
10 GRAPHICS 2
20 POKE 756,226
30 SETCOLOR 0,0,0
35 FOR X=0 TO 10
40 POSITION X,5:PRINT #6;" "
50 POSITION X,6:PRINT #6;"lettres "
52 POSITION X,7:PRINT #6;"minuscules "
60 POSITION X,8:PRINT #6;" "
65 FOR TEMPS=1 TO 50:NEXT TEMPS
70 NEXT X
```

Vous êtes arrivé au questionnaire final de cette cassette. La piste sonore n'est pas utilisée sur cette dernière partie.

PRECAUTIONS D'EMPLOI

Le logiciel que vous venez d'acquérir, se présente sous l'une des trois formes suivantes : cartouche, cassette ou disquette.

Si le programme est en cartouche, ne l'exposez pas à une chaleur excessive ni à une atmosphère corrosive ou des conditions atmosphériques extrêmes qui pourraient endommager les contacts.

Si le programme est en cassette ou en disquette, respectez bien les conseils ci-après :

- Ne l'exposez pas à la chaleur : évitez de poser une disquette sur l'ordinateur ou sur l'unité de disquette.
- Ne touchez jamais avec les doigts la bande magnétique elle-même ou la disquette magnétique
- Ne fumez pas ! En effet, les poussières très fines composant la fumée se déposent sur les parties magnétiques et il s'ensuit une usure prématurée du support.
- Veillez à tenir ces supports suffisamment éloignés de tout champ magnétique. Ne posez pas de disquette sur votre téléviseur ou à proximité de l'écran.
- N'éteignez jamais votre unité de disquette alors que la lampe « BUSY » est allumée. Vous risquez de magnétiser les têtes.
- Toute opération d'allumage ou d'extinction de l'unité de disquette doit se faire sans disquette à l'intérieur.
- N'écrivez jamais directement sur la disquette.
- N'utilisez jamais d'épingle ou de trombone sur une disquette.
- Stockez de préférence vos disquettes et vos cassettes à la verticale.
- Evitez toute contrainte mécanique sur ces supports ; ne posez pas de livre par-dessus, ne les faites pas tomber par terre, etc.
- Ne faites pas d'encoche, de perforation ou de découpe dans la pochette de la disquette.

En pratique, il est facile de respecter toutes ces conditions, mais n'oubliez pas que 98 % des défauts constatés proviennent en fait d'une mauvaise manipulation.

